
#include <IRremote.h>     //Library V4.7
#include <avr/sleep.h>
#include <avr/wdt.h>

const int RECV_PIN = 2;   // IR Receiver connected to Pin 2
const unsigned long SLEEP_COMMAND = 0xED126B86;  //POWER ON/OFF ON LCD REMOTE
const int AWAKE_PIN = 12; // Status LED for awake/sleep intervals

void setup() {
  Serial.begin(9600);
  IrReceiver.begin(RECV_PIN, ENABLE_LED_FEEDBACK);
  pinMode(LED_BUILTIN, OUTPUT);
  pinMode(AWAKE_PIN, OUTPUT);
  digitalWrite(AWAKE_PIN,HIGH);
}

void loop() {
  if (IrReceiver.decode()) {
    unsigned long command = IrReceiver.decodedIRData.decodedRawData;
    Serial.print("Received: ");
    Serial.println(command, HEX);

    if (command == SLEEP_COMMAND) {
      digitalWrite(AWAKE_PIN,LOW);
      Serial.println("Going to sleep...");
      delay(100); // Allow serial to flush
      enterSleep();
    }
    IrReceiver.resume(); 
  }
}
// Watchdog Interrupt Service Routine (ISR)
ISR(WDT_vect) {
  // This is called when the WDT "bites"
  // It wakes the MCU and continues execution after sleep_mode()
  digitalWrite(AWAKE_PIN,HIGH);
}
void enterSleep() {
  set_sleep_mode(SLEEP_MODE_PWR_DOWN); // Deepest sleep
  
  // Configure Watchdog Timer for 8-second interval
  cli(); // Disable interrupts
  wdt_reset();
  
  WDTCSR |= (1 << WDCE) | (1 << WDE);
  // Set interrupt mode (WDIE) and 8s timeout (WDP3 and WDP0)
  WDTCSR = (1 << WDIE) | (1 << WDP3) | (1 << WDP0);
  sei(); // Enable interrupts

  sleep_enable();
  digitalWrite(LED_BUILTIN, LOW); // Turn off LED to save power
  sleep_mode();                   // CPU stops here until WDT or IR interrupt

  // --- Wakes up here after WDT timeout ---
  sleep_disable();
  wdt_disable(); // Stop WDT after waking
  Serial.println("Woke up!");
}

